
Abstract. This article presents an approach of meaningful rare sequen-
tial pattern mining based on the declarative programming paradigm of
Answer Set Programming (ASP). The setting of rare sequential pattern
mining is introduced. Our ASP approach provides an easy manner to
encode expert constraints on expected patterns to cope with the huge
amount of meaningless rare patterns. Encodings are presented and quan-
titatively compared to a procedural baseline. An application on care
pathways analysis illustrates the interest of expert constraints encoding.

1 Introduction

Pattern mining aims at extracting meaningful structured knowledge hidden in
large amounts of raw data. Building on the hypothesis that an interesting pattern
occurs frequently in the data, most research on pattern mining have focused
on frequent pattern mining. However, in many cases, rare patterns can also be
meaningful. For example, this is true when physicians want to identify dangerous
outcomes out of ordinary procedures from care pathway data. Fortunately for
patients, such outcomes are rares but, unfortunately for data analysts, such
events are di�cult to extract using standard approaches.

Mining rare patterns has been studied in the context of itemsets [1]. But
to the best of our knowledge, the problem of mining rare sequential patterns
has not been addressed yet. The lack of work on this topic has been recently
identi�ed by Hu et al. [2] as an important matter for future research.

The main problem with rare patterns, known from experiments on rare item-
sets, is their huge number. Condensed rare patterns, called �minimal rare pat-
terns� [3], have been proposed to reduce the number of patterns to extract with-
out loosing information. Yet, it is desirable to further reduce the number of
patterns and to improve the pattern signi�cance. The approach we propose in
this paper, is to let the expert express extra constraints to specify the most in-
teresting patterns. To achieve this goal, we need to develop a method to extract
condensed rare sequential patterns which will be versatile enough to support
easy addition of expert constraints. However, most approaches based on proce-
dural programs would require speci�c and long developments to integrate extra

constraints. Declarative programming appears to be an interesting alternative
to extract knowledge from data under complex expert constraints.

Using declarative programming, and more especially logic programming ap-
pears to be an interesting alternative to dedicated procedural algorithm several
decades ago and received a new recent interests with the improvement of solver
e�ciency. The state-of-the-art can be organized according to two analysis axis:
the data analysis tasks and the declarative programming paradigm. Data anal-
ysis task are classically separated in supervised vs unsupervised learning. More
especially, inductive logic programming (ILP) [4] belongs to the �eld of super-
vised machine learning, while the more recent �eld of declarative pattern mining
[5], which consists in mining patterns with declarative encodings, belongs to the
�eld of unsupervised learning. For these two tasks, several declarative paradigms
have been proposed: logic programming, Prolog or Answer Set Programming
(ASP), Constraint Programming (CP) and Satis�ability Solving (SAT). For each
approach, the objective is to propose a uniform representation technique for ex-
amples, background knowledge and hypotheses. ILP started with encodings in
Prolog [4], and alternative systems have been implemented based on CP [6] or
on ASP [7]. In declarative pattern mining, most of the approaches have been
implemented using SAT [8], CP [9,10] and some with ASP [11]. In this work, we
choose the paradigm of ASP to implement our rare sequential pattern mining to
bene�t from a �rst-order logic programming language. It makes the encoding of
extra constraints easier and its solvers have proved their e�ciency [11].

The contributions of this paper are threefold: (i) we formalize the general
problem of rare sequential pattern mining and we model it in ASP, together
with two important variations of this problem. Thanks to the �exibility of ASP,
we were able to solve all three problems with the same ASP solver, avoiding the
tedious work of designing and implementing an algorithm for each new problem.
(ii) We provide important insights on modelling e�ciency by comparing several
alternative models. The experimental comparison shows that general ASP-based
approaches can compete with ad-hoc specialized algorithms. And 3), we apply
rare sequential mining to our target application of analyzing care pathway data.
In particular, we demonstrate the bene�t of additional constraints to extract
meaningful results in this domain.

2 Rare sequential patterns: de�nitions and properties

This section introduces the basic concepts of rare sequential pattern mining.
Let I = {i1, i2, . . . , i|I|} be a set of items. A sequence s, denoted by 〈sj〉1≤j≤n

is an ordered list of items sj ∈ I.
Given two sequences s = 〈sj〉1≤j≤n and s′ = 〈s′j〉1≤j≤m with n ≤ m, we

say that s is a subsequence of s′, denoted s � s′, i� there exists n integers
i1 < . . . < in such that sk = s′ik , ∀k ∈ {1, . . . , n}.

Let D =
{
s1, s2, . . . , sN

}
, be a dataset of N sequences. The support of any

sequence s, denoted supp(s), is the number of sequences si ∈ D that contain s:

supp(s) =
∣∣{si ∈ D|s � si

}∣∣ (1)



Mining rare sequential patterns with ASP 3

s1 s2 s3 s4 s5 s6 s7

Seq. 〈d, a, b, c〉 〈a, c, b, c〉 〈a, b, c〉 〈a, b, c〉 〈a, c〉 〈b〉 〈c〉
Table 1. Example of arti�cial dataset of 7 sequences.

Let σ ∈ N+ be a frequency threshold given by the data analyst. A sequence
s is rare i� supp(s) < σ. In such case, the sequence s is called a rare sequential
pattern or a rare pattern for short.

It is noteworthy that rare patterns are strongly related to the frequent pat-
terns, i.e. the sequence s such that supp(s) ≥ σ. The set of rare patterns is the
dual of frequent patterns. A pattern that is not rare is frequent and conversely.

Moreover, the rarity constraint is monotone, i.e. for two sequences s and
s′, if s � s′ and s is rare, then s′ is rare. This property comes from the anti-
monotonicity of the support measure.

Example 1 (Sequences and rare patterns). Table 1 presents a simple dataset D,
de�ned over a set of items I = {a, b, c, d}. The support of sequence p1 = 〈a, b〉
is 4 and the support of p2 = 〈c, c〉 is 1. For a support threshold σ = 2, p2 is a
rare pattern but p1 is not. Due to the monotonicity of rarity, every extension of
p2 is also rare (e.g. 〈a, c, c〉).

We now introduce two additional de�nitions: zero-rare patterns and minimal
rare patterns. A zero-rare pattern is a rare pattern with a null support. This
means that this pattern does not occur in dataset sequences. A non-zero-rare
pattern is a rare pattern which is not zero-rare.

Finally, a sequence s is a minimal rare pattern (mRP) i� it is rare but all its
proper subsequences are frequent.

mRP = {s|supp(s) < σ ∧ ∀s′ ≺ s, supp(s′) ≥ σ} (2)

Example 2 (minimal rare patterns). Figure 1 illustrates the lattice of non-zero
rare patterns in the dataset D of Table 1 for σ = 2. 〈b, a〉 (not �gured out in the
lattice) is an example of zero-rare pattern insofar as it does not appear in D.
〈a, c, b〉 is a non-zero-rare pattern. However, it is not an mRP since 〈c, b〉, one of
its subpatterns (i.e. 〈c, b〉 ≺ 〈a, c, b〉), is rare.

3 Mining rare sequential patterns with ASP

In this section, we detail ASP encodings for mining rare and minimal rare se-
quential patterns. ASP is a declarative programming paradigm founded on the
semantic of stable models [12]. From a general point of view, declarative pro-
gramming gives a description of what is a problem instead of specifying how to
solve it. A program describes the problem and a dedicated solver �nds out its
solutions.

ASP provides a high-level syntax that makes the program almost easy to
understand compared to other declarative paradigms. An ASP program is a set

{}

< a > < b > < c > < d >

< a, b > < a, c > < b, c > < c, b > < c, c > < d, a > < d, b > < d, c >

< a, b, c > < a, c, b > < a, c, c > < c, b, c > < d, a, b > < d, b, c >

FP

RP

mRP

< d, a, b, c > < a, c, b, c >

Fig. 1. Lattice of sequential patterns of dataset D. Zero-rare patterns are omitted for
sake of clarity. In white: frequent patterns, in grey: rare patterns, surrounded: minimal
rare patterns for σ = 2.

seq(1,1,d). seq(1,2,a). seq(1,3,b). seq(1,4,c).

seq(2,1,a). seq(2,2,c). seq(2,3,b). seq(2,4,c).

seq(3,1,a). seq(3,2,b). seq(3,3,c).

seq(4,1,a). seq(4,2,b). seq(4,3,c).

seq(5,1,a). seq(5,2,c).

seq(6,1,b).

seq(7,1,c).

Listing 1.1. Facts specifying the sequence dataset of Table 1

of rules of the form a0 :- a1, . . . , am, not am+1, . . . , not an, where each ai is a
propositional atom for 0 ≤ i ≤ n and not stands for default negation. Atoms
can be encoded using a �rst-order logic syntax. If n = 0, such rule is called a
fact. If a0 is omitted, such rule represents an integrity constraint. The reader can
refer to Janhunen et al. [12] for a more detailed introduction to ASP. Dedicated
solvers, such as clingo [13], ensure the e�cient solving of ASP encodings.

In the following, Section 3.1 introduces the ASP generation of candidate
sequential patterns, then the rarity constraints is introduced in Section 3.2 and
�nally, Section 3.3 introduces additional constraints to extract e�ciently only
the minimal-rare patterns.

First of all, the dataset has to be encoded as facts. It is represented by
atoms seq(s,t,i) stating that item i occurs at time t in sequence s. Listing 1.1
illustrates facts encoding of the sequence dataset of Table 1.

3.1 Enumerate candidate sequential patterns in ASP

Listing 1.2 speci�es the candidate generation. It reuses some notations and prin-
ciples introduced by Gebser et al. [11]. Line 1 lists all atoms present in the
dataset. For each item i ∈ I a unique atoms item(i) is generated. The token "_"
stands for an anonymous variable that does not recur anywhere. Line 2 de�nes
the sequence lengths of each sequence from the dataset. The length of the se-
quence is an item position for which there is no items afterwards. Lines 4 to 7
enumerate all possible patterns. A pattern is a sequence of positions to �ll up



Mining rare sequential patterns with ASP 5

with exactly one item4. It is encoded by atoms pat(x,i) where x is a position
in the pattern and i ∈ I. Predicate patpos/1 takes one variable and de�nes the
sequence position for pattern items. Rules with curly brackets in the head are
choice rules such as lines 5 and 7. Rule at line 5 makes the pattern length choice
which can not be larger than maxlen. Line 7 chooses exactly one atom pat/2 for
each position X and thus generates the candidate patterns combinatorics.

Lines 9 to 13 evaluate the pattern support. This encoding uses the �ll-gaps
strategy proposed in [11] (see Figure 2). The idea consists in mapping each
sequence position to one (and exactly one) pattern position. In addition, some
sequence positions are associated with an item that does not match the item in
the mapped pattern position. This denotes �lling positions and indicates which
part of the pattern has already been �read �. Lines 9 to 11 traverse each sequence
s = 〈sj〉1≤j≤n in D to yield atoms of the form occ(s,l,pl) where pl is a position
in the pattern and l is a position in the sequence s. occ(s,l,pl) indicates that
the pl-th �rst items of the patterns have been read at position l of the sequence.
Line 9 searches for a compatible position P with the �rst pattern item. Line 11
yields a new atom occ(s,l,pl) while the (l−1)-th item has been read at position
(pl − 1) and the pl-th sequence item matches the l-th pattern item. In any case,
line 10 yields an atom associating position pl with already read positions of the
pattern. The support(s) is yielded line 13 while sequence t holds the current
pattern.

1 item(I) :- seq(_, _,I).

2 seqlen(S,L) :- seq(S,L,_), not seq(S,L+1,_).

4 patpos (1).

5 { patpos(X+1) } :- patpos(X), X<maxlen.

6 patlen(L) :- patpos(L), not patpos(L+1).

7 1 { pat(X,I) : item(I) } 1 :- patpos(X).

9 occ(S,1,P) :- pat(1,I), seq(S,P,I).

10 occ(S,L,P) :- occ(S, L, P-1), seq(S,P,_).

11 occ(S,L,P) :- occ(S, L-1, P-1), seq(S,P,C), pat(L,C).

13 support(S) :- occ(S, L, LS), patlen(L), seqlen(S,LS).

Listing 1.2. Encoding of candidate patterns generation

For more details about �ll-gaps strategy and its alternatives, we refer the
readers to the article of Gebser et al. [11].

3.2 E�cient encoding of rarity constraint (ASP-RSM)

The modularity of ASP encoding enables to easily add constraints to the previous
program. Where Gebser et al. added constraints to select only the frequent
patterns, we select the answer sets that correspond to non-zero-rare patterns.

4 We do not consider sequences of itemsets for sake of encoding simpli�cation.

Fig. 2. Illustration of the �ll-gaps embedding strategy (see Listing 1.2) for pattern
〈abc〉 in the sequence 〈. . . a . . . b . . . bdc . . . b . . . 〉. Each black-circle illustrates one occ/3
atoms in a model. ASP atoms of this predicate are detailed for three circles.

The following constraint states that the number of supported sequences must
be strictly below the given threshold th (i.e. not above th). th is a program
constant that can be speci�ed while the solver is launched.

:- #count{ S : support(S) } >= th.

Despite the correctness of this constraint, we propose an alternative encoding
that explicitly expresses the monotonicity property such that the solver may
prune non-zero-rare patterns more e�ciently. This encoding introduces a rare/1

predicate (see listing 1.3). Instead of evaluating the support of pattern 〈pj〉1≤j≤n,
it evaluates independently the support of each of its pre�xes. For any l, 1 ≤ l ≤
n, rare(l) means that the subpattern 〈pj〉1≤j≤l is rare. Line 18 imposes to have
the atom rare(n) where n is the pattern length. Line 17 gives a simpler manner
to yield such atom: if a pre�x-pattern of length l is rare, then all patterns of
length l′, l′ > l, are rare. This rule prevents the solver from evaluating all rules
of line 15 and it potentially prevents from evaluating occ/3 atoms. Finally, line 19
prunes zero-rare patterns imposing to have at least one supported sequence.

15 rare(IL):- IL=1..L, patlen(L),

16 #count{ S : occ(S, IL , LS),seqlen(S,LS) } < th.

17 rare(L) :- rare(L-1), L<=PL , patlen(PL).

18 :- not rare(L), patlen(L).

19 :- not support(S) : seq(S,_,_).

Listing 1.3. Encoding of rare sequence mining

As the number of non-zero-rare patterns may be very high, the following
section presents ASP-MRSM, an extension of ASP-RSM that extracts the con-
densed set of minimal-rare patterns, mRP.

3.3 ASP Minimal Rare Sequence Miner (ASP-MRSM)

The encoding in Listing 1.4 can be used to mine minimal rare sequences. It is
based on the observation that it is su�cient to evaluate the support of all sub-
pattern of size n−1 to determine whether a pattern p of size n is a minimal. Let
p = 〈pi〉1≤i≤n be a non-zero-rare pattern. p is a mRP i� ∀u ∈ {1, . . . , n}, the

Fig. 3. mRP occurences for pattern p = 〈a, b, c, d〉 and u = 3, on the left, in case of
a sequence that support pu but not p. Grey disks illustrate occ/3 atoms and black
circles illustrate spocc/4.

sub-pattern pu = 〈pi〉i∈{1,...,u−1,u+1,...,n} is frequent (i.e. not rare). According
to Equation 2, p is a mRP i� ∀p′ ≺ p, p′ is frequent. It is then clear that if p is
a mRP, then ∀u, pu is frequent. Conversely, for all sequence s such that s ≺ p,
∃u such that s ≺ pu. Then, as pu is frequent and by anti-monotonicity of the
frequency constraint, s is frequent too.

Furthermore, the encoding takes advantage of two properties of sequen-
tial patterns: (1) a sequence that supports the pattern p supports each pat-
tern pu, (2) if a sequence s is supported by pu but not by p, then for all
l ∈ [1, u − 1], occ(s,l,pl) is part of the ASP model computed by Listing 1.2,
where (pl)l∈[1,(n−1)] is an occurrence of pu. In fact, as there is no di�erence
between pu and p before position u, there is no di�erences between their occur-
rences. These properties are helpful to decide whether pu is rare. In Listing 1.4,
line 20 enumerates all subpatterns, identi�ed by U. Similarly to occ/3 predicates,
atoms spocc(s,u,l,pl) describe the occurrence of pu in sequence s (Figure 3).

Lines 22 to 26 compute occurrences of pattern pu according to the �ll-gaps
principle (see section 3.1) with two di�erences: (1) spocc/4 are yielded only
if sequence t is not supported, otherwise the sequence is obviously supported
knowing that pu ≺ p; and (2) spocc/4 starts from pattern position u using
occ(t,u− 1,_) available atoms to avoid redundant computation according to
the second property above. Lines 28 to 33 determine whether the subpattern u
is rare. This encoding di�ers from section 3.2 mainly at lines 30-32. The total
number of sequences supported by pu is the number of sequences supported by
p (line 32) plus the number of sequences supported by pu but not by p (line 31),
i.e. sequences for which spocc/4 atoms reached the last pattern item. Finally,
line 33 eliminates answer sets for which a subpattern is rare.

It is worth noticing that in ASP two answer sets can not share information
during the solving process. Gebser et al. [11] proposed to use a solver extension,
called asprin, to encode the mining of closed patterns, i.e. patterns which have
any larger patterns with the same support. A similar approach may be used for
mRP. Our approach prefers a pure declarative but more e�cient encoding.

4 Experiments and results

In this section, we evaluate our approach on synthetic and real datasets. First,
we use synthetic datasets to compare performances of our ASP encodings with

20 suppat(U) :- U=1..L, patlen(L), L>1.

22 spocc(S,1,2,P) :- seq(S,P,I), pat(2,I), not support(S).

23 spocc(S,U,U,P) :- suppat(U), occ(S,U-1,P), not support(S).

24 spocc(S,U,L ,P+1) :- spocc(S,U,L,P), seq(S,P+1,_).

25 spocc(S,U,L+1,P+1) :- spocc(S,U,L,P), seq(S,P+1,I),

26 pat(L+1,I).

28 sprare(U,U-1):- sprare(U-1), suppat(U).

29 sprare(U,L) :- sprare(U, L-1), L<=LP, patlen(LP).

30 sprare(U, IL):- suppat(U), IL=U+1..L, patlen(L),

31 #count{ S: spocc(S,U,IL ,LS), seqlen(S,LS);

32 S: support(S) } < th.

33 :- sprare(U,L), patlen(L).

Listing 1.4. Encoding part for minimal rare patterns

procedural algorithms. Then, we demonstrate the �exibility of our declarative
pattern mining approach on a case study.

In all experiments, we used clingo, version 5.0. All programs were run on a
computing server with 16Gb RAM. Multi-threading capabilities of clingo have
been disabled because existing procedural algorithms are not parallel and be-
cause multi-threading introduces high variance in run times measurements.

4.1 Runtime and memory e�ciency on synthetic datasets

In this �rst set of experiments, we use synthetic datasets to compare the e�ciency
of our ASP encodings with procedural algorithms. Four approaches have been
compared: ASP-RSM and Apriori-Rare extract non-zero-rare patterns; and ASP-
MRSM and MRG-EXP extract minimal rare patterns. Since Apriori-Rare [3]
and MRG-Exp [14], were originally developed for rare itemset mining, they were
adapted to mine rare sequential patterns4.

We use several synthetic datasets to measure the impact of various param-
eters such as vocabulary size and means sequence length on the runtime and on
the memory usage. Our sequence generator4 produces datasets following a 3-
step retro-engineering procedure inspired from the IBM Quest Synthetic Data
Generator: 1) a set of random patterns is generated, 2) for each pattern, a list
of occurrences is generated, and 3) synthetic sequences are built so that each
sequence contains a pattern if the sequence is in the pattern occurrence list. The
sequences are then completed with random items so that the mean sequence
length of the resulting dataset reaches the input parameter λ.

4 The source codes of the ASP programs, of procedural algorithms and of the dataset
generator are available at http://www.ilp-paper-1.co.nf/

http://www.ilp-paper-1.co.nf/

Fig. 4. (a) is computing times w.r.t. threshold for synthetic databases of size 105 (with
a mean sequence length of 15). (b) and (c) are mean computing times w.r.t. sequence
length and vocabulary size (with |D| = 250 and σ = 20%).

Thres-
hold

#patterns Memory usage (bytes)

(σ) #Rare #mRP ASP-RSM ASP-MRSM Apriori-Rare MRG-EXP
Dataset � 1000 sequences

5% 1 011 117 260 1.0×108 0.7×108 2.5×106 5.1×106
10% 1 011 993 527 1.0×108 0.7×108 5.3×106 7.5×106
20% 1 012 360 2 280 1.1×108 0.7×108 9.6×106 1.2×107
30% 1 012 428 7 123 1.3×108 0.9×108 9.8×106 5.3×107

Dataset � 5000 sequences

5% 105 907 477 1.0×108 3.5×108 5.2×105 5.8×105
10% 109 861 957 1.0×108 3.5×108 5.2×106 7.3×105
20% 110 879 4 754 1.0×108 3.7×108 5.6×106 3.7×106
30% 111 022 16 530 1.2×108 4.8×108 7.8×106 3.8×107

Dataset � 10000 sequences

5% 108 266 456 2.0×108 6.5×108 1.3×106 1.3×106
10% 110 334 734 2.0×108 6.5×108 3.4×106 3.1×106
20% 110 955 4 086 2.0×108 6.5×108 5.6×106 3.6×106
30% 111 036 10 177 2.0×108 8.0×108 7.2×106 7.3×106

Table 2. Memory consumption and number of patterns of ASP-RSM, ASP-MRSM,
Apriori-Rare and MRG-EXP w.r.t dataset size and rarity threshold (with λ = 10).

Figure 4 presents runtimes for variable dataset size (a), variable sequence
length (b), and variable vocabulary size (c). Table 2 presents the number of
extracted patterns and the memory footprint.

First, we can see in Figure 4-(a) that ASP-MRSM is an order of magnitude
faster than alternatives approaches. This is an important result since previous
work on declarative pattern mining have shown that declarative approaches are
often slower than the procedural counterparts [11,15]. This experiment demon-
strates that the overhead induced by the declarative encoding is balanced by the
more e�cient solving strategy achieved by the ASP solver.

Second, we focus on ASP-RSM and Apriori-Rare (non-zero-rare-patterns). In
general, decreasing the threshold induces a larger search space and thus higher
runtimes. However, we can see that ASP-RSM is far less sensitive to variations
of the threshold than Apriori-Rare is. As a consequence, ASP-RSM is faster for
very low thresholds. Apriori-Rare is greatly impacted by the increasing search
space, even though the number of rare patterns remains almost constant as it

can be seen in Table 2. In contrast the runtime of ASP-RSM remains fairly low,
even when the threshold is low. This demonstrates again that the ASP solver is
able to explore the search space more e�ciently.

Figure 4-(b) shows that both declarative and procedural approaches are pe-
nalised by larger sequence lengths. The algorithm behaviour on this plot is con-
sistent with previous plots and ASP-MRSM remains the most e�cient approach.

Figure 4-(c) shows the mean computing time with an increasing vocabulary
size |I|. Understanding the impact of the size of the vocabulary is not trivial,
since a smaller vocabulary means fewer items to process but also increases the
dataset density. In Table 2, we can see that the fewer are items, the fewer are
rare patterns (with a �xed size, |D| and mean length, λ). Remark that the slope
decreases with the vocabulary size: the number of patterns increases quickly for
small vocabularies, but increases slowly for larger vocabularies. This behaviour
is explained by the combinatorics of items: with a lot of rare items, the mean
length of mRP is smaller because it is more probable to have patterns of length
1 that are non-zero rare (with our simulated dataset). In fact, the shorter is an
mRP, the more rare patterns it represents. As a consequence, the computation
time increases less faster.

4.2 Application to rare care pathways analytics

We now apply our approach to care pathway analytics to illustrate that the
number of rare patterns can be reduced using additional expert constraints. This
is not a new result but we show that extracting rare patterns becomes practically
interesting with constraints. The ASP approach was here crucial to incorporate
constraints easily. Care pathway analytics is a research �eld that consists in
exploring patient care pathways to understand care uses, more especially drugs
consumption in this case, and their impacts on patients health.

This case study analyzes care pathways of patients exposed to anti-epileptic
drugs who had epileptic seizures. Rare pattern mining aims at identifying (i)
patients to exclude from the study because of their unexpected rare care path-
ways, for instance due to speci�c heavy treatments (cancer, hepatitis C, AIDS,
etc.); (ii) rare adverse drugs reactions. The dataset is made of 500 care pathways
with drugs deliveries during one year. This represents a total amount of 101, 793
events, with |I| = 3, 671. Drug deliveries are encoded with seq(t,p,e) atoms
where e is the active component of the delivered drug.

The rough mining of minimal rare sequential patterns with σ = 10% yields
7, 758 mRPs of length at most 3 (maxlen = 3). This pattern number makes
these results di�cult to analyze by clinicians.

Two types of constraints have been added in order to illustrate the pattern
reduction. The �rst one is a relaxation of rare patterns. Too rare patterns are
poorly signi�cant but so numerous. We thus introduced a second parameters
fmin such that the support of extracted patterns must be between fmin and
σ. The second one concerns the maximum duration between two events of a
pattern occurrence. This constraint, known as the maxgap constraint, avoids to

Fig. 5. Number of rare patterns extracted from care pathways w.r.t expert constraints.

enumerate an occurrence while two of its events are too far in time. It assumes
that the �rst event may causally be related to the second one.

Figure 5 illustrates the e�ciency of this constraint to reduce the number of
rare patterns with σ = 10% and maxlen = 3. The plot on the left illustrates the
number of patterns w.r.t. the maxgap constraints value (from 3 to 25 events).
Considering that having larger max-gap reduces the number of non-zero pat-
terns, this number increases with maxgap. The plot on the right illustrates the
number of patterns w.r.t. the minimal frequency constraint, fmin (from 0.4%
to 6%). The closer fmin is to σ, the less the number of patterns. This num-
ber decreases exponentially with fmin. The number of patterns extracted with
fmin = 6% fall down to 87. This number of patterns is su�ciently low to be pre-
sented to clinicians. Moreover, the computing time is reduced by using expert
constraints. This �nal experiment is solved in 425s which is signi�cantly faster
than in our experiments on synthetic datasets.

The two constraints can be combined. We choose to select patterns with
fmin = 5% and maxgap = 3. In such case, the number of patterns fall down
to 133. One interesting pattern describes the deliveries of lamotrigine, an anti-
epileptic drug, co-occurring with two antibiotics ceftriaxone and amoxicillin.
This unexpected potential adverse drug reaction may lead to new epidemiological
questions.

5 Conclusion

This article extends existing work, notably Gebser et al. [11] work, on declarative
sequential pattern mining with Answer Set Programming by proposing e�cient
encoding of rare patterns. It shows the versatility of declarative approach to de-
sign easily a new data mining task. We shown, on the one hand, that the ASP
encodings appears to be an e�cient solution compared to procedural approaches,
but that its memory consumption limits the analysis to mid-size databases. This
problem is known in pattern mining with ASP [11] and the solution will be to
use dedicated propagators [15] to improve computing e�ciency of the approach.
On the other hand, we illustrate that this approach answers an applicative ques-
tion by extracting sequential patterns satisfying additional expert constraints.
We shown that constraints prune e�ciently useless patterns with bene�t on
computation time.

